Dynamics of Machines Week 4 –1st and 2nd Exercises

Connecting Rod – Physical Pendulum



$$J_{AZ}\ \ddot{\chi} = -l_{SA}\ m_r\ g\ \sin\chi$$

$$\ddot{\chi} = -\frac{l_{SA}\ m_r\ g}{J_{AZ}} \sin\chi \quad (rad/s^2) \quad angular\ \ acceleration$$

$$l_{SA} = 0{,}15\ m \qquad J_{AZ} = 0{,}05\ kgm^2$$

$$m_r = 0{,}45\ kg \qquad g = 10\ m/s^2$$

Initial Conditions:

$$\chi_0 = 0{,}3\ rad - initial\ \ angle$$

$$\dot{\chi}_0 = 0{,}5\ rad/s - initial\ \ angular\ \ velocity$$



4/1 Exercise (Runge-Kutta method)

```
// week 4 – 1st exercise
// Solve the following second order ordinary differential equation with Runge-Kutta method
    //                          Jaz*(y")=-lSA*mr*g*sin(y)
    // Rearrangement:           y"=-(lSA*mr*g/Jaz)*sin(y)
    // Initial conditions:
    //              t=0 (s) ➔ y(0)=0.3 (rad) angle
    //              t=0 (s) ➔ y'(0)=0.5 (rad/s) angular velocity
    //---------------------------------------------------------------------------------------------------------------
    // Let's define a p column matrix. It has two rows which contains the following data:
    //              p(1)=y (rad) angle
    //              p(2)=y' (rad/s) angular velocity
    // Derivative of the p matrix: p' , where p'(1)=y' so the same as p(2) (m/s) angular velocity
    //              p'(1)=y' so p(2) (rad/s) angular velocity
    //              p'(2)=y" (rad/s^2) angular acceleration
    //              p'(2)=y"=-(lSA*mr*g/Jaz)*sin(y)
    //              p'(2)=y"=-(lSA*mr*g/Jaz)*sin(p1)
clear;
//------------------Variables:  ------------------------------------------------------------------------------------
lSA=0.15;           // length between S and A points (m)
mr=0.45;            // mass of the connecting rod (kg)
JAZ=0.05;           // moment of inertia of the connecting rod about z axis (kgm^2)
g=10.0;             // gravitational acceleration (m/s^2)
// Initial Conditions //////////
t0=0;               // initial time
p0(1)=0.3           // initial angle y(0)= 0.3  (rad)  ( = 17.191 degree)
p0(2)=0.5;          // initial angular velocity, y'(0)= 0.5 (rad/s)
// Time interval
t=0:0.001:10;       // from 0s to 10s with 0,001s time increment
//-------------------------------------------------------------------------------------------------------------------
function [pdot]=f(t, p)
    // Let's define the elements of p' matrix
    pdot(1)=p(2)                            // angular velocity
    pdot(2)=-(lSA*mr*g/JAZ)*sin(p(1))       // rearranged function
endfunction
//
//use ode command to solve the differential equation ---------------------------------------------------------------
p=ode("rk",p0,t0,t,f); // rk - means Runge-Kutta method, p0 – initial values matrix, t0 – initial time, t – time
                         interval, f – righ hand side of the differential equation
    ////////////////////////  Caution !!!  ///////////////////
    // With this method we can not get the angular acceleration function we just get the p matrix where p(1,:)
    contains the values of the angle function and p(2,:) contains the values of the angular velocity function
    // So we have to calculate the angular acceleration function with the values of these two functions
a=-(lSA*mr*g/JAZ)*sin(p(1,:));                  // angular acceleration  (rad/s^2)
//
// Plotting results ---------------------------------------------------------------------------------------
subplot(3,1,1)        //Divide the graphic window into 3x1 matrix of sub-windows with subplot command
plot2d(t,p(1,:)*(180/(%pi)),1);                 // plot the angle funciton in degree unit
xtitle(" Angle function","time (s)","angle (degree)");
xgrid(2);
    //
```

```
subplot(3,1,2)
plot2d(t,p(2,:),5);
xtitle("Angular Velocity function","time (s)","angular velocity (rad/s)");
xgrid(2);
    //
subplot(3,1,3)
plot2d(t,a(1,:),1);
xtitle("Angular Acceleration function ","time (s)"," angular acceleration (rad/s^2)");
xgrid(2);
```

>>>>>> Save the script.

Execute: Execute >>>> …. file with no echo



4/2 Exercise (Euler method)

```
// week 4 – 2nd exercise
        // Solve the following second order ordinary differential equation with Euler method
        //                                JAZ*(y'')= -lSA*mr*g*sin(y)
        // Rearrangement:                 y''=-(lSA*mr*g/JAZ)*sin(y)
        // Initial conditions:
        //                                t=0 (s) ➔ y(0)=0.3 (rad) angle
        //                                t=0 (s) ➔ y'(0)=0.5 (rad/s) angular velocity
clear;
//------------------Variables -------------------------------------------------------------------
lSA=0.15;          // length between S and A point (m)
mr=0.45;           // mass of the connecting rod (kg)
JAZ=0.05;          // moment of inertia of the connecting rod about z axis (kgm^2)
g=10.0;            // gravitational acceleration (m/s^2)
//  Time interval
dt=0.001;                    // time increment (s)
tmax=10.0;                   // final time value of the calculation (s)
n=int(tmax/dt);              // number of time steps
//  Initial Conditions ////////////
t0=0;                        // initial time (s)
y0=0.3;                      // initial angle y(0)= 0.3  (rad)  ( = 17.191 degree)
v0=0.5;                               // initial angular velocity, y'(0)= 0.5 (rad/s)
a0=-(lSA*mr*g/JAZ)*sin(y0);           // initial angular acceleration
// -----------------------------------------------------------------------------------------------
t=(1:n);
y=(1:n);
v=(1:n);
a=(1:n);
// 1st elements of the matricies   //////  i=1 loop variable ///////////
t(1)=0;
y(1)=y0;
v(1)=v0;
a(1)=a0;
/////////// for loop /// Numerical Integration ///////
for i=2:n
   t(i)=t0+dt;
   a(i)=-(lSA*mr*g/JAZ)*sin(y0);
   v(i)=v0+((a(i)+a0)/2)*dt;          // Trapezoidal Rule
   y(i)=y0+((v(i)+v0)/2)*dt;          // Trapezoidal Rule
        /// Variable value exchange:
   t0=t(i);
   y0=y(i);
   v0=v(i);
   a0=a(i);
 end
//
// Plotting results //////////////////////
subplot(3,1,1)  //Divide the graphic window into 3x1 matrix of sub-windows with subplot command
plot2d(t,y(1,:)*(180/(%pi)),1);                 // plot the angle funciton in degree unit
xtitle(" Angle function","time (s)","angle (degree)");
xgrid(2);
       //
```

```
subplot(3,1,2)
plot2d(t,v(1,:),5);
xtitle("Angular Velocity function","time (s)","angular velocity (rad/s)");
xgrid(2);
    //
subplot(3,1,3)
plot2d(t,a(1,:),1);
xtitle("Angular Acceleration function ","time (s)"," angular acceleration (rad/s^2)");
xgrid(2);
```

>>>>>> Save the script.

Execute: Execute >>>> …. file with no echo